Never trust a CAS!

- **PolynomialQ**

```
PolynomialQ[1 / x, {x, y}]
```

```
False
```

```
PolynomialQ[1 / x, {y}]
```

```
True
```

```
PolynomialQ[1 / x, {}]
```

```
False
```

```
PolynomialQ[1 / x]
```

```
False
```

```
PolynomialQ[Sin[x], {x, y}]
```

```
False
```

```
PolynomialQ[Sin[x], {y}]
```

```
True
```

```
PolynomialQ[Sin[x], {}]
```

```
True
```

```
PolynomialQ[Sin[x]]
```

```
True
```

```
PolynomialQ[Sqrt[x], {}]
```

```
True
```

```
PolynomialQ[Sqrt[x^2 - 1], {}]
```

```
False
```

- **Chebyshev polynomials**

```
Table[ChebyshevU[n, -1], {n, -5, 5}]
```

```
{-4, 3, 2, -1, 0, 1, -2, 3, -4, 5, -6}
```

```
Table[(-1)^n * (n + 1), {n, -5, 5}]
```

```
{4, -3, 2, -1, 0, 1, -2, 3, -4, 5, -6}
```

Limit

```
Limit[x^k * Binomial[-2, k], k → Infinity, Assumptions → Abs[x] < 1]
```

```
ComplexInfinity
```

```
FullSimplify[Binomial[-2, k]]
```

```
FullSimplify::infd : Expression Binomial[-2, k] simplified to ComplexInfinity. >>
```

```
ComplexInfinity
```

Binomial sum

```
Sum[(-1)^(i + j) Binomial[i + j, i] Binomial[m, i] Binomial[n, j], {i, 0, m}, {j, 0, n}]
```

$$\sum_{i=0}^{m}\sum_{j=0}^{n}(-1)^{i+j}\,\text{Binomial}[i+j,i]\,\text{Binomial}[m,i]\,\text{Binomial}[n,j]$$

```
Sum[(-1)^(i + j) Binomial[i + j, i] Binomial[n, i] Binomial[n, j], {i, 0, n}, {j, 0, n}]
```

```
1/(-1 + n) (1 - DifferenceRoot[Function[{ẏ, ṅ}, {-(-1 + ṅ) (1 + ṅ) (-ṅ + n) ẏ[ṅ] + (ṅ - ṅ^2 - 2 ṅ^3 - n + 2 ṅ^2 n) ẏ[
        1 + ṅ] + ṅ^2 (1 + ṅ - n) ẏ[2 + ṅ] == 0, ẏ[1] == 0, ẏ[2] == -1/(1 - n)}]][0] +
    n DifferenceRoot[Function[{ẏ, ṅ}, {-(-1 + ṅ) (1 + ṅ) (-ṅ + n) ẏ[ṅ] + (ṅ - ṅ^2 - 2 ṅ^3 - n + 2 ṅ^2 n)
        ẏ[1 + ṅ] + ṅ^2 (1 + ṅ - n) ẏ[2 + ṅ] == 0, ẏ[1] == 0, ẏ[2] == -1/(1 - n)}]][0] +
    DifferenceRoot[Function[{ẏ, ṅ}, {-(-1 + ṅ) (1 + ṅ) (-ṅ + n) ẏ[ṅ] + (ṅ - ṅ^2 - 2 ṅ^3 - n + 2 ṅ^2 n)
        ẏ[1 + ṅ] + ṅ^2 (1 + ṅ - n) ẏ[2 + ṅ] == 0, ẏ[1] == 0, ẏ[2] == -1/(1 - n)}]][1 + n] -
    n DifferenceRoot[Function[{ẏ, ṅ}, {-(-1 + ṅ) (1 + ṅ) (-ṅ + n) ẏ[ṅ] + (ṅ - ṅ^2 - 2 ṅ^3 - n + 2 ṅ^2 n)
        ẏ[1 + ṅ] + ṅ^2 (1 + ṅ - n) ẏ[2 + ṅ] == 0, ẏ[1] == 0, ẏ[2] == -1/(1 - n)}]][1 + n])
```

Limits of Zeta functions

```
ass = n > 1 && Element[n, Integers] && a > 0 && b > 0 && k < n
```

```
n > 1 && n ∈ Integers && a > 0 && b > 0 && k < n
```

```
expr1 = x^k * Zeta[n, a + b * x];
expr2 = x^k * Zeta[n, 1 + a + b * x];
```

```
Timing[Limit[expr1, x → Infinity, Assumptions → ass]]
```

```
{0.736046, 0}
```

```
Timing[Limit[expr2, x → Infinity, Assumptions → ass]]
```

```
{0.676043, 0}
```

```
Timing[Limit[expr1 - expr2, x → Infinity, Assumptions → ass]]
```

```
{53.8954, Limit[x^k Zeta[n, a + b x] - x^k Zeta[n, 1 + a + b x],
  x → ∞, Assumptions → n > 1 && n ∈ Integers && a > 0 && b > 0 && k < n]}
```

- **StruveL**

```
N[StruveL[1 + I, 700 * I], 25]
```

```
-0.2056171312291138282112197 + 0.05092642840654207727239512 ⅈ
```

```
N[StruveL[1 + I, 100 * I], 25]
```

```
0.1745249349140313153158107 + 0.08029354364282519308755572 ⅈ
```

```
Plot[Re[StruveL[1 + I, x * I]], {x, 0, 1000}]
```




```
Plot[Re[StruveL[1 + I, x * I]], {x, 480, 520}]
```



Keep control of your expressions!

PolynomialGCD

```
MyPolynomialGCD[p1_, p2_, x_] :=
  Module[{e1, e2},
   If[p1 * p2 === 0, Return[p1 + p2]];
   While[p2 =!= 0,
    While[(e1 = Exponent[p1, x]) ≥ (e2 = Exponent[p2, x]),
     p1 = p1 - (Coefficient[p1, x, e1] / Coefficient[p2, x, e2]) * x^(e1 - e2) * p2;
    ];
    {p1, p2} = {p2, p1};
   ];
   Return[p1];
  ];
```

```
MyPolynomialGCD[x^2 - 1, x - 1, x]
```

```
Set::write : Tag Plus in -1 + x^2 is Protected. >>

Set::write : Tag Plus in -1 + x^2 is Protected. >>

Set::write : Tag Plus in -1 + x^2 is Protected. >>

General::stop :
 Further output of Set::write will be suppressed during this calculation. >>

$Aborted
```

```
MyPolynomialGCD[pp1_, pp2_, x_] :=
  Module[{p1 = pp1, p2 = pp2, e1, e2},
   If[p1 * p2 === 0, Return[p1 + p2]];
   While[p2 =!= 0,
    While[(e1 = Exponent[p1, x]) ≥ (e2 = Exponent[p2, x]),
     p1 = p1 - (Coefficient[p1, x, e1] / Coefficient[p2, x, e2]) * x^(e1 - e2) * p2;
    ];
    {p1, p2} = {p2, p1};
   ];
   Return[p1];
  ];
```

```
MyPolynomialGCD[x^2 - 1, x - 1, x]

∞::indet : Indeterminate expression x^-∞ encountered. >>

                                   1
Power::infy : Infinite expression  — encountered. >>
                                   0

∞::indet : Indeterminate expression x^∞ encountered. >>

Coefficient::ivar : 1 is not a valid variable. >>

∞::indet : Indeterminate expression x^-∞ encountered. >>

General::stop :
 Further output of ∞::indet will be suppressed during this calculation. >>

                                   1
Power::infy : Infinite expression  — encountered. >>
                                   0

Coefficient::ivar : 1 is not a valid variable. >>

                                   1
Power::infy : Infinite expression  — encountered. >>
                                   0

General::stop :
 Further output of Power::infy will be suppressed during this calculation. >>

Coefficient::ivar : 1 is not a valid variable. >>

General::stop :
 Further output of Coefficient::ivar will be suppressed during this calculation. >>

$Aborted

MyPolynomialGCD[pp1_, pp2_, x_] :=
  Module[{p1 = pp1, p2 = pp2, e1, e2},
   If[p1 * p2 === 0, Return[p1 + p2]];
   While[p2 =!= 0,
    While[(e1 = Exponent[p1, x]) ≥ (e2 = Exponent[p2, x]),
     p1 = Expand[p1 - (Coefficient[p1, x, e1] / Coefficient[p2, x, e2]) * x^(e1 - e2) * p2];
    ];
    {p1, p2} = {p2, p1};
   ];
   Return[p1];
  ];

MyPolynomialGCD[x^2 - 1, x - 1, x]

-1 + x

MyPolynomialGCD[x^2 - 1, (y - 1) x - 1, x]

$Aborted

MyPolynomialGCD[pp1_, pp2_, x_] :=
  Module[{p1 = pp1, p2 = pp2, e1, e2},
   If[p1 * p2 === 0, Return[p1 + p2]];
   While[p2 =!= 0,
    While[(e1 = Exponent[p1, x]) ≥ (e2 = Exponent[p2, x]),
     p1 = Together[p1 - (Coefficient[p1, x, e1] / Coefficient[p2, x, e2]) * x^(e1 - e2) * p2];
    ];
    {p1, p2} = {p2, p1};
   ];
   Return[p1];
  ];
```

```
MyPolynomialGCD[x^2 - 1, (y - 1) x - 1, x]
```

$$\frac{2\,y - y^2}{(-1 + y)^2}$$

- **NullSpace**

```
mat = Table[x - RandomInteger[10], {9}, {10}]
```

```
{{-4 + x, -10 + x, -4 + x, -2 + x, -7 + x, -6 + x, -1 + x, -8 + x, -5 + x, -10 + x},
 {-1 + x, -3 + x, -9 + x, -2 + x, -10 + x, -2 + x, -1 + x, -4 + x, -6 + x, -5 + x},
 {-4 + x, -10 + x, -10 + x, -5 + x, -10 + x, -9 + x, -9 + x, -10 + x, -7 + x, -1 + x},
 {-7 + x, -3 + x, -9 + x, -4 + x, -3 + x, -3 + x, x, -8 + x, -10 + x, -1 + x},
 {-7 + x, -8 + x, -1 + x, -5 + x, -10 + x, -9 + x, -2 + x, -5 + x, -10 + x, -2 + x},
 {-5 + x, -1 + x, -3 + x, -4 + x, -2 + x, -3 + x, -7 + x, -3 + x, -1 + x, -4 + x},
 {-5 + x, -4 + x, -3 + x, -5 + x, -1 + x, -10 + x, x, -9 + x, -3 + x, -9 + x},
 {-10 + x, -6 + x, -9 + x, x, -4 + x, -4 + x, -1 + x, x, -3 + x, x},
 {-8 + x, -5 + x, -5 + x, -1 + x, -8 + x, -4 + x, -10 + x, x, -10 + x, -8 + x}}
```

```
Timing[ns = NullSpace[mat];]
```

{4.10026, Null}

```
ByteCount[ns]
```

2 124 540 576

- **Vandermonde Determinant**

```
mat = Table[1 / (x + i) ^j, {i, 0, 7}, {j, 0, 7}];
```

```
Timing[det = Det[mat];]
```

{0.216013, Null}

```
ByteCount[det]
```

16 524 064

```
Timing[Together[det]]
```

$$\left\{28.6898,\ \frac{125\,411\,328\,000}{x^7\,(1+x)^7\,(2+x)^7\,(3+x)^7\,(4+x)^7\,(5+x)^7\,(6+x)^7\,(7+x)^7}\right\}$$

```
ByteCount[det]
```

16 524 064

Know what's behind!

```
Exponent[x^2 + x + 1 - x * (x - 1), x]
```

1

```
Timing[Exponent[Product[x + y + z + i, {i, 100}], x]]
```

{61.5838, 100}

```
Denominator[1 / (x + 1) - x / (x - 1)]
```

1

```
Denominator[x^(-n)]
```

x^n

Internal representation

```
Max[Cases[x / 20 + y / 21, _Integer, Infinity]]
```

$-\infty$

```
A = Table[1 / (i + j - 1), {i, 3}, {j, 3}]
```

$$\left\{\left\{1, \frac{1}{2}, \frac{1}{3}\right\}, \left\{\frac{1}{2}, \frac{1}{3}, \frac{1}{4}\right\}, \left\{\frac{1}{3}, \frac{1}{4}, \frac{1}{5}\right\}\right\}$$

```
MatchQ[A, {{(_Rational) ..} ..}]
```

False

```
ser = Series[Log[1 - x], {x, 0, 10}]
```

$$-x - \frac{x^2}{2} - \frac{x^3}{3} - \frac{x^4}{4} - \frac{x^5}{5} - \frac{x^6}{6} - \frac{x^7}{7} - \frac{x^8}{8} - \frac{x^9}{9} - \frac{x^{10}}{10} + O[x]^{11}$$

```
ser - Last[ser]
```

$$-1 - x - \frac{x^2}{2} - \frac{x^3}{3} - \frac{x^4}{4} - \frac{x^5}{5} - \frac{x^6}{6} - \frac{x^7}{7} - \frac{x^8}{8} - \frac{x^9}{9} - \frac{x^{10}}{10} + O[x]^{11}$$

Never use Simplify

Simplify

```
Simplify[x^4 + 4 (x^3 + x) + 6 x^2 + 1]
```

$(1 + x)^4$

```
Simplify[(x^2 + 1) (x^2 - 1)]
```

$-1 + x^4$

A rational function

```
RandomPolynomial[x_, d_] := Sum[RandomInteger[100] * x^i, {i, 0, d}]
```

```
expr = Sum[RandomPolynomial[x, 5] / RandomPolynomial[x, 5], {50}];
```

```
Timing[Expand[expr];]
```

{0.004, Null}

```
Timing[Together[expr];]
```

```
{0.348022, Null}
```

```
Timing[Simplify[expr];]
```

```
{1.24408, Null}
```

```
Timing[FullSimplify[expr];]
```

```
{153.174, Null}
```

▪ Trigonometric functions

```
Simplify[Sin[x]^2 + Cos[x]^2]
```

```
1
```

```
Expand[Sin[x]^2 + Cos[x]^2, Trig → True]
```

```
1
```

▪ Factorials

```
Simplify[(n + 1)! / n!]
```

$$\frac{(1+n)!}{n!}$$

```
FunctionExpand[(n + 1)! / n!]
```

```
1 + n
```

```
FullSimplify[(n + 1)! / n!]
```

```
1 + n
```

▪ Logical expressions

```
Simplify[x && (x || y)]
```

```
x
```

```
LogicalExpand[x && (x || y)]
```

```
x
```